# Limit of a real function

Evaluating

Limit[2^x Log[x], x -> 0, Direction -> -1]


(right limit) gives correctly the answer

-∞


but the command

Limit[2^x Log[x], x -> 0, Direction -> 1]


(left limit) gives again

-∞


The same in WolframAlpha. What is the reason behind this ?

-

You are correct in that the limit of the complex logarithm on the standard branch, as $x\to 0$ from below, should better be written as $-\infty + i \pi$.

The reason why Mathematica doesn't give this result is that it uses a polar representation for infinity in the complex plane. You can see this by doing

FullForm[Limit[2^x Log[x], x -> 0, Direction -> 1]]


DirectedInfinity[-1]

This DirectedInfinity expresses the fact that the imaginary part goes to a constant whereas the real part diverges.

However, in complex analysis it is often necessary to retain constants such as the one in $-\infty + i \pi$, e.g., to choose integration paths correctly. If you want to get that constant from a Limit, I think the easiest way to do it is to forcibly do the limits of the real and imaginary parts separately:

Table[
Limit[h[2^x Log[x]], x -> 0, Direction -> 1], {h, {Re, Im}}]

(* ==> {-Infinity, Pi} *)

-

It's a correct result.It is best to illustrate on the plot.

f[x_] := 2^x*Log[x];
{Limit[f[x], x -> 0, Direction -> 1],
Limit[f[x], x -> 0, Direction -> -1]}


$\{-\infty ,-\infty \}$

Plot[{Re@f[x], f[x]}, {x, -2, 2}, Filling -> Bottom,
PlotRange -> {{-2, 2}, {-10, 3}}, PlotLegends -> "Expressions"]

-

It's a correct result. The computation below may help to see this.

Table[2^x Log[x], {x, -(10.^Range[-1, -10, -1])}]

(* Out[48]= {-2.14838785758 + 2.93120959177 I, -4.57335995192 +
3.119892088 I, -6.90296884693 + 3.13941582202 I, -9.20970198196 +
3.14137490253 I, -11.5128456637 + 3.1415708778 I, -13.8155009818 +
3.141590476 I, -16.1180945337 + 3.14159243583 I, -18.4206806163 +
3.14159263181 I, -20.7232658226 + 3.14159265141 I, -23.0258509283 +
3.14159265337 I} *)


The values are heading to negative infinity, on a path near Pi units north of the negative axis. Remark: for better or worse, Mathematica does not have a notion of "infinite real part plus finite imaginary part", so -Infinity+I*Pi just becomes -Infinity.

-